9.3 Application Fourier Series of Piecewise Smooth Functions (p608)

To define piecewise functions you can use either piecewise (do ?piecewise for the syntax) or use the book's suggestion, which is to define a function "unit" that is one on the interval (a,b) and zero otherwise:

```
> unit := (t,a,b) -> Heaviside(t-a) - Heaviside(t-b);
```

$$unit := (t, a, b) \rightarrow \text{Heaviside}(t-a) - \text{Heaviside}(t-b) \qquad (1)$$

```
> plot(unit(t,0.5,1.5),t=-2..2);
```



Here is the function in Figure 9.3.8. This function is even, thus to find its Fourier coefficients we only need to define it on [0,Pi]:

```
> f := t -> (Pi/3)*unit(t, 0, Pi/6) + (Pi/2 - t)*unit(t, Pi/6, 5*Pi/6) + (-Pi/3)*unit(t, 5*Pi/6, Pi);
```

$$f := t \rightarrow \frac{1}{3}\pi\, unit\left(t, 0, \frac{1}{6}\pi\right) + \left(\frac{1}{2}\pi - t\right) unit\left(t, \frac{1}{6}\pi, \frac{5}{6}\pi\right) - \frac{1}{3}\pi\, unit\left(t, \frac{5}{6}\pi, \pi\right) \qquad (2)$$

To double check that we have the correct expression we can plot the function

```
> plot(f(t),t=0..Pi);
```



Since the function is EVEN b_n = 0 for n>=1, thus we only compute the a_n:

```
> a := n -> (2/Pi) * int(f(t)*cos(n*t),t=0..Pi);
```

$$a := n \rightarrow \frac{2\left(\int_0^{\pi} f(t)\cos(n\,t)\,\mathrm{d}t\right)}{\pi} \tag{3}$$

The "assuming integer" at the end of the expression tells Maple that n is an integer for addition simplifications, however an expression does not come to mind!

```
> a(n) assuming integer;
```

$$\frac{2\left(\cos\left(\frac{1}{6}\,n\,\pi\right) - \cos\left(\frac{5}{6}\,n\,\pi\right)\right)}{\pi\,n^2} \tag{4}$$

So we simply compute the Fourier sum, with say 25 terms

```
> fouriersum := a(0)/2 + sum(a(n)*cos(n*t),n=1..25);
```

$$\mathit{fouriersum} := \frac{2\cos(t)\sqrt{3}}{\pi} - \frac{2}{25}\,\frac{\sqrt{3}\,\cos(5\,t)}{\pi} - \frac{2}{49}\,\frac{\sqrt{3}\,\cos(7\,t)}{\pi} \tag{5}$$

$$+\frac{2}{121}\frac{\sqrt{3}\cos(11\,t)}{\pi}+\frac{2}{169}\frac{\sqrt{3}\cos(13\,t)}{\pi}-\frac{2}{289}\frac{\sqrt{3}\cos(17\,t)}{\pi}$$

$$-\frac{2}{361}\frac{\sqrt{3}\cos(19\,t)}{\pi}+\frac{2}{529}\frac{\sqrt{3}\cos(23\,t)}{\pi}+\frac{2}{625}\frac{\sqrt{3}\cos(25\,t)}{\pi}$$

And we check that the Fourier sum is consistent with the function we asked.

```
> plot(fouriersum, t= -2*Pi..3*Pi);
```



In the same way it is possible to compute the coefficients b(n) for odd functions or more generally the a(n) and b(n) for general (not odd neither even) functions.